

Calcolo dei coefficienti per i metodi multistep

Enrico Bertolazzi

[-] Carica le librerie

```
> restart ;  
with(plots):  
Warning, the name changecoords has been redefined
```

[-] La procedura **AB** calcola i pesi per una formula Adams-Bashforth a N+1 passi

```
> AB := proc(N::integer)  
  local X, Y, g, p, ip, x ;  
  # costruisce la successione di punti per il polinomio interpolante  
  X := [seq(k, k=-N..0)] ;  
  Y := [seq(g(k), k=-N..0)] ;  
  # costruisce il polinomio interpolante  
  p := interp( X, Y, x ) ;  
  # calcola l'integrale del polinomio interpolante  
  ip := int(p, x=0..1);  
  # la successione dei pesi e`  
  return seq(coeff(ip,g(k),1),k=-N..0) ;  
end proc ;
```

```
AB := proc(N::integer)  
local X, Y, g, p, ip, x;  
  X := [seq(k, k = -N .. 0)];  
  Y := [seq(g(k), k = -N .. 0)];  
  p := interp(X, Y, x);  
  ip := int(p, x = 0 .. 1);  
  return seq(coeff(ip, g(k), 1), k = -N .. 0);  
end proc;
```

[-] La procedura **AM** calcola i pesi per una formula Adams-Multon a N+1 passi

```
> AM := proc(N)  
  local X, Y, g, p, ip, x ;  
  # costruisce la successione di punti per il polinomio interpolante  
  X := [seq(k, k=-N..1)] ;  
  Y := [seq(g(k), k=-N..1)] ;  
  # costruisce il polinomio interpolante  
  p := interp( X, Y, x ) ;  
  # calcola l'integrale del polinomio interpolante
```

```

ip := int(p, x=0..1);
# la successione dei pesi e`
return seq(coeff(ip,g(k),1),k=-N..1) ;
end proc ;

```

```

AM := proc(N)
local X, Y, g, p, ip, x;
X := [seq(k, k = -N .. 1)];
Y := [seq(g(k), k = -N .. 1)];
p := interp (X, Y, x);
ip := int(p, x = 0 .. 1);
return seq(coeff(ip, g(k), 1), k = -N .. 1);
end proc;

```

```
> # esempio metodo di Eulero esplicito
```

```
AB(0) ;
```

1

```
> # esempio metodo dei Trapezi
```

```
AM(0) ;
```

$\frac{1}{2}, \frac{1}{2}$

```
> # esempio metodo esplicito a 4 passi
```

```
AB(3) ;
```

$-\frac{3}{8}, \frac{37}{24}, -\frac{59}{24}, \frac{55}{24}$

```
> # tabella degli Adams-Bashforth
```

```
AB(0) ;
```

```
AB(1) ;
```

```
AB(2) ;
```

```
AB(3) ;
```

```
AB(4) ;
```

```
AB(5) ;
```

```
AB(6) ;
```

```
AB(7) ;
```

```
AB(8) ;
```

1

$-\frac{1}{2}, \frac{3}{2}$

$\frac{5}{12}, -\frac{4}{3}, \frac{23}{12}$

$-\frac{3}{8}, \frac{37}{24}, -\frac{59}{24}, \frac{55}{24}$

$\frac{251}{720}, -\frac{637}{360}, \frac{109}{30}, -\frac{1387}{360}, \frac{1901}{720}$

$$\frac{-95}{288}, \frac{959}{480}, \frac{-3649}{720}, \frac{4991}{720}, \frac{-2641}{480}, \frac{4277}{1440}$$

$$\frac{19087}{60480}, \frac{-5603}{2520}, \frac{135713}{20160}, \frac{-10754}{945}, \frac{235183}{20160}, \frac{-18637}{2520}, \frac{198721}{60480}$$

$$\frac{-5257}{17280}, \frac{32863}{13440}, \frac{-115747}{13440}, \frac{2102243}{120960}, \frac{-296053}{13440}, \frac{242653}{13440}, \frac{-1152169}{120960}, \frac{16083}{4480}$$

$$\frac{1070017}{3628800}, \frac{-4832053}{1814400}, \frac{19416743}{1814400}, \frac{-45586321}{1814400}, \frac{862303}{22680}, \frac{-69927631}{1814400}, \frac{47738393}{1814400}, \frac{-21562603}{1814400}, \frac{14097247}{3628800}$$

> # tabella degli Adams-Moulton

- AM(0) ;
- AM(1) ;
- AM(2) ;
- AM(3) ;
- AM(4) ;
- AM(5) ;
- AM(6) ;
- AM(7) ;
- AM(8) ;

$$\frac{1}{2}, \frac{1}{2}$$

$$\frac{-1}{12}, \frac{2}{3}, \frac{5}{12}$$

$$\frac{1}{24}, \frac{-5}{24}, \frac{19}{24}, \frac{3}{8}$$

$$\frac{-19}{720}, \frac{53}{360}, \frac{-11}{30}, \frac{323}{360}, \frac{251}{720}$$

$$\frac{3}{160}, \frac{-173}{1440}, \frac{241}{720}, \frac{-133}{240}, \frac{1427}{1440}, \frac{95}{288}$$

$$\frac{-863}{60480}, \frac{263}{2520}, \frac{-6737}{20160}, \frac{586}{945}, \frac{-15487}{20160}, \frac{2713}{2520}, \frac{19087}{60480}$$

$$\frac{275}{24192}, \frac{-11351}{120960}, \frac{1537}{4480}, \frac{-88547}{120960}, \frac{123133}{120960}, \frac{-4511}{4480}, \frac{139849}{120960}, \frac{5257}{17280}$$

$$\frac{-33953}{3628800}, \frac{156437}{1814400}, \frac{-645607}{1814400}, \frac{1573169}{1814400}, \frac{-31457}{22680}, \frac{2797679}{1814400}, \frac{-2302297}{1814400}, \frac{2233547}{1814400}, \frac{1070017}{3628800}$$

$$\frac{8183}{1036800}, \frac{-116687}{1451520}, \frac{335983}{907200}, \frac{-462127}{453600}, \frac{6755041}{3628800}, \frac{-8641823}{3628800}, \frac{200029}{90720}, \frac{-1408913}{907200}, \frac{9449717}{7257600}, \frac{25713}{89600}$$

>